COMPITO di ARCHITETTURA DEI CALCOLATORI del 22-06-2021

MATRICOLA _____

COGNOME _____

NOME _____

DA RESTITUIRE INSIEME AGLI ELABORATI e A TUTTI I FOGLI
→ NON USARE FOGLI NON TIMBRATI
→ ANDARE IN BAGNO PRIMA DELL'INIZIO DELLA PROVA
→ NO FOGLI PERSONALI, NO TELEFONI, SMARTPHONE/WATCH, ETC

NOTA: dovrà essere consegnato l'elaborato dell'es.1 come file <COGNOME>.s e quelli dell'es. 4 come files <COGNOME>.v e <COGNOME>.png

1) [10/30] Trovare il codice assembly RISC-V corrispondente al seguente micro-benchmark (**utilizzando solo e unicamente istruzioni dalla tabella sottostante**), rispettando le convenzioni di uso dei registri dell'assembly (riportate qua sotto, per riferimento).
Nota: la funzione `sqrt` e' un'istruzione (`fsqrt.s`) che estrae la radice quadrata.

```
float *hada1(int l) {
    int i, j, k, t; float m = sqrt(l);
    float *H = (float *)sbrk(l*l*sizeof(float));
    for (i = 0; i < l; i++)
        for (j = 0; j < l; j++) {
            if (i == 0 && j == 0) H[i*l+j] = 1 / m;
            else for (k = 0; k <= 10; k++) {
                t = 1<<k;
                if (i >= t && j < t)
                    H[i*l+j] = H[(i - t)*l + j];
            }
        }
    return H;
}
```

```
int main() {
    float *H, s = 0; int k;
    H = hada1(4);
    for (k = 0; k < 16; ++k) s += H[k];
    print_float(s);
    exit(0);
}
```

RISCV Instructions (RV64IMFD) **v210622**

Instruction coding (hexadecimal) opcode+funct3+{funct7,imm}	Instruction	Example		Register operation	Meaning (** instructions available only in RV64, i.e. 64-bit case)
33+0+00/3b+0+00	**add**	**add/addw**	**x5,x6,x7**	x5 ← x6 + x7	Add two operands; exception possible (addw**)
33+0+20/3b+0+20	**subtract**	**sub/subw**	**x5,x6,x7**	x5 ← x6 – x7	Subtracts two operands; exception possible (subw**)
13+0+imm/1b+0+imm	**add immediate**	**addi/addiw**	**x5,x6,100**	x5 ← x6 + 100	Add a constant ; exception possible (addiw**)
33+0+01/3b+0+01	**multiply**	**mul/mulw**	**x5,x6, x7**	x5 ← x6 * x7	(signed/word) Lower 64 bits of 128-bits product (mulw**)
33+01+01	**multiply high**	**mulh**	**x5,x6,x7**	x5 ← x6 * x7	Higher 64bits of 128-bits product
33+4+01/3b+4+01	**division**	**div/divw**	**x5,x6,x7**	x5 ← x6/x7	(signed/word) division (divw**)
33+6+01/3b+6+01	**reminder**	**rem/remw**	**x5,x6,x7**	x5 ← x6 % x7	Reminder of the division (remw**)
33+2+0/33+3+0	**set on less than**	**slt/sltu**	**x5,x6,x7**	if (x6 < x7) x5 ← 1; else x5 ← 0	(signed/unsigned) compare x6 and x7 (less than)
13+2+imm/13+3+imm	**set on less than immediate**	**slti/sltiu**	**x5,x6,100**	if (x6 < 100) x5← 1; else x5 ← 0	(signed/unsigned) compare x6 and 100 (less than)
33+7+0/33+6+0/33+4+0	**and / or / xor**	**and/or/xor**	**x5,x6,x7**	x5← x6&x7 / x6|x7 / x6^ x7	Logical AND/OR/XOR
13+7+imm/13+6+imm/13+4+imm	**and /or / xor immediate**	**andi/ori/xori**	**x5,x6,100**	x5 ← x6&100 / x6|100 / x6^100	Logical AND/OR/XOR register, constant
33+1+0/3b+1+0	**shift left logical**	**sll/sllw**	**x5,x6,x7**	x5 ← x6 << x7	Shift left by register (sllw**)
13+1+imm/1b+1+imm	**shift left logical immediate**	**slli/slliw**	**x5,x6,10**	x5 ← x6 << 10	Shift left by the immediate value (slliw**)
33+5+0/3b+5+0	**shift right logical**	**srl/srlw**	**x5,x6,x7**	x5 ← x6 >> x7	Shift right by register (srlw**)
13+5+imm/1b+5+imm	**shift right logical immediate**	**srli/srliw**	**x5,x6,10**	x5 ← x6 >> 10	Shift left by immediate value (srliw**)
33+5+20/3b+5+20	**shift right arithmetic**	**sra/sraw**	**x5,x6,x7**	x5 ← x6 >> x7 (arith.)	Shift right by register (sign is preserved) (sraw**)
13+5+imm/1b+5+imm	**shift right arithmetic immediate**	**srai/sraiw**	**x5,x6,10**	x5 ← x6 >> 10 (arith.)	Shift right by immediate value (sraiw**)
03+3+imm/03+2+imm/03+0+imm	**load dword / word / byte**	**ld/lw/lb**	**x5,100(x6)**	x5 ← MEM[x6+100]	Data from memory to register
03+6+imm/03+4+imm	**load word / byte unsigned**	**lwu/bu**	**x5,100(x6)**	x5 ← MEM[x6+100]	Data from mem. To reg.; no sign extension (lwu**)
23+3+imm/23+2+imm/23+0+imm	**store dword / word / byte**	**sd/sw/sb**	**x5,100(x6)**	MEM[x6+100] ← x5	Data from register to memory (sw**)
37+imm[31:12] (no funct3)	**load upper immediate**	**lui**	**x5,0x12345**	x5 ← 0x1234'5000	Load most significant 20 bits
PSEUDOINSTRUCTION	**load address**	**la**	**x5,var**	x5 ← &var **(PSEUDO INST.)** load address of 'var' in x5	**REAL: lui x5,H20(&var);ori x5, L12(&var) INST.** (H20=high 20 bit of &var; L12=low 12 bits of &var)
6f+imm[31:12](rd=0) 63+0+imm[11:0](rs1=rs2=0)	**jump/branch**	**j/b**	**label**	PC+=off (off=PC-&label) **(PS.INST.)**	**REAL INST.: jal x0,offset/beq x0,x0,offset**
6f+0+imm[31:12](rd=1,no funct3)	**jump and link (offset)**	**jal**	**label**	x1←(PC+4);PC+=offset **(PS. INST.)**	**REAL INST.: jal x1,offset** (offset=PC-&label)
67+0+imm (rd=0,rs1=1)	**return from procedure**	**Ret**		PC←x1 **(PSEUDO INST.)**	**REAL INST.: jalr x0,0(x1)**
67+0+imm	**jump and link register**	**jalr**	**x1, 100(x5)**	x1 ← (PC + 4);PC=x5+100	Procedure return; indirect call
63+0+(imm÷2)/63+1+(imm÷2)	**branch on equal / not-equal**	**beq/bne**	**x5,x6,100**	if (x5 = =/!= x6) PC=PC+100	Equal / Not-equal test; PC relative branch
73+0+0 (rs1=0,rs2=0,rd=0)	**ecall**	**Ecall**		SEPC←PC;PC←STVEC;save PL/IE;PL=1;IE=0	Call OS (service number in a7); PL= privilege lev; IE=int.en.
73+0+8 (rs1=0,rs2=2,rd=0)	**sret**	**Sret**		PC←SEPC; restore PL/IE	Exit supervisor mode; PL= privilege lev; IE=int.en.
PSEUDOINSTRUCTION	**move**	**mv**	**x5,x6**	x5 ← x6 **(PSEUDO INST.)**	**REAL INST.: add x5,x0,x6**
PSEUDOINSTRUCTION	**load immediate**	**li**	**x5,100**	x5 ← 100 **(PSEUDO INST.)**	**REAL INST.: addi x5,x0,100**
PSEUDOINSTRUCTION	**no operation (nop)**	**nop**		do nothing **(PSEUDO INST.)**	**REAL INST.: addi x0,x0,0**
53+0+{0,1}/53+0+{4,5}	**floating point add/sub**	**fadd/fsub.{s,d}**	**f0,f1,f2**	f0←f1+f2 / f0←f1-f2	Single or double precision add / subtract
53+0+{8,9}/53+0+{c,d}	**floating point multiplication/division**	**fmul/fdiv.{s,d}**	**f0,f1,f2**	f0←f1*f2 / f0←f1/f2	Single or double precision multiplication / division
PSEUDOINSTRUCTION	**floating point move between f-regs**	**fmv.{s,d}**	**f0,f1**	f0←f1 **(PSEUDO INST.)**	**REAL INST.: fsgnj.{s,d} f0,f1,f1**
PSEUDOINSTRUCTION	**floating point negate**	**fneg.{s,d}**	**f0,f1**	f0← – (f1) **(PSEUDO INST.)**	**REAL INST.: fsgnjn.{s,d} f0,f1,f1**
PSEUDOINSTRUCTION	**floating point absolute value**	**fabs.{s,d}**	**f0,f1**	f0← | f1 | **(PSEUDO INST.)**	**REAL INST.: fsgnjx.{s,d} f0,f1,f1**
53+0/1/2+{50,51}	**floating point compare**	**fle/flt/feq.{s,d}**	**x5,f0,f1**	x5← (f0<f1)	Single and double: compare f0 and f1 <=,<,==
53+0+{70,71} (rs2=0)	**move between x (integer) and f regs**	**fmv.x.{s,d}**	**x5,f0**	x5←f0 (no conversion)	Copy (no conversion)
53+0+{78,79} (rs2=0)	**move between f and x regs**	**fmv.{s,d}.x**	**f0,x5**	f0←x5 (no conversion)	Copy (no conversion)
7+2+imm/27+2+imm	**load/store floating point (32bit)**	**flw/fsw**	**f0,0(x5)**	f0←MEM[x5] / MEM[x5]←f0	Data from FP register to memory
7+3+imm/27+3+imm	**load/store floating point (64bit)**	**fld/fsd**	**f0,0(x5)**	f0←MEM[x5] / MEM[x5]←f0	Data from FP register to memory
53+7+21(rs2=0)/53+7+20(rs2=0)	**convert to/from double from/to single**	**fcvt.d.s/fcvt.s.d**	**f0,f1**	f0← (double)f1 / f0← (single)f1	Type conversion
53+7+{60,61} (rs2=0)	**convert to integer from {single,double}**	**fcvt.w.{s,d}**	**x5,f0**	x5← (int)f0	Type conversion
53+7+{68,69} (rs2=0)	**convert to {single,double} from integer**	**fcvt.{s,d}.w**	**f0,x5**	f0← ({single,double})x5	Type conversion
53+0+{2c,2d} (rs2=0)	**square root**	**fsqrt.{s,d}**	**f0,f1**	f0← square root of f1	Single or double square root
53+0/1/2+{10,11}	**sign injection**	**fsgnj/jn/jx.{s,d}**	**f0,f1,f2**	f0←sgn(f2)|f1| /–sgn(f2)|f1| /sgn(f2)f1|	Extract the mantissa and exp. from f1 and sign from f2

Register Usage

Register	ABI Name	Usage
x10-x11	**a0-a1**	arguments and results
x9, x18-x27	**s1, s2-s11**	Saved
x5-7, x28-x31	**t0-t2, t3-t6**	Temporaries
x12-x17	**a2-a7**	Arguments

Register	ABI Name	Usage
x0	**zero**	The constant value 0
x8, x2	**s0/fp, sp**	frame pointer, stack pointer
x1, x3	**ra, gp**	return address, global pointer
x4	**tp**	thread pointer

Register	ABI Name	Usage
f10-f11	**fa0-fa1**	Argument and Return values
f8-f9, f18-f27	**fs0-fs1, fs2-fs11**	Saved registers
f0 – f7, f28-f31	**ft0-ft7, ft8-ft11**	Temporaries registers
f12-17	**fa2-fa7**	Function arguments

System calls

Service Name	Serv.No.(a7)	INPUT Arguments	OUTPUT Args
print_int	1	a0=integer to print	---
print_float	2	fa0=float to print	---
print_double	3	fa0=double to print	---
print_string	4	a0=address of ASCIIZ string to print	---
read_int	5	---	a0=integer

Service Name	Serv.No.(a7)	INPUT Arguments	OUTPUT Arguments
read_float	6	---	fa0=float
read_double	7	---	fa0=double
read_string	8	a0=address of input buffer, a1=max chars to read	---
sbrk	9	a0=Number of bytes to be allocated	a0=pointer to allocated memory
exit	10	---	---

COMPITO di ARCHITETTURA DEI CALCOLATORI del 22-06-2021

2) [6/30] Si consideri una cache di dimensione 64B e a 2 vie di tipo write-back/write-non-allocate. La dimensione del blocco e' 8 byte, il tempo di accesso alla cache e' 4 ns e la penalita' in caso di miss e' pari a 40 ns, la politica di rimpiazzamento e' LRU. Il processore effettua i seguenti accessi in cache, ad indirizzi al byte: 125, 170, 167, 245, 183, 119, 235, 163, 288, 309, 310, 308, 213, 196, 377, 166, 362, 233, 163, 169. Tali accessi sono alternativamente letture e scritture. Per la sequenza data, ricavare il tempo medio di accesso alla cache, riportare i tag contenuti in cache al termine, i bit di modifica (se presenti) e la lista dei blocchi (ovvero il loro indirizzo) via via eliminati durante il rimpiazzamento ed inoltre in corrispondenza di quale riferimento il blocco e' eliminato.

3) [5/30] Spiegare con proprie parole il funzionamento del "Modo 4" del timer 8254, il cui diagramma temporale è riportato in figura. Inoltre, indicare con precisione: i) il significato dei segnali rappresentati in tale diagramma, ii) come deve essere impostata la parola di controllo CWR e il relativo registro di conteggio per ottenere questo diagramma supponendo di utilizzare N=64000, il contatore n.2 in conteggio binario.



4) [9/30] Descrivere e sintetizzare in Verilog una rete sequenziale utilizzando il modello di Mealy-Ritardato con un ingresso X su due bit e un'uscita Z su un bit tale che se X[1] e X[0] sono entrambi 1 e X[0] ha assunto per tre volte il valore uno nei cicli precedenti non necessariamente consecutivi, l'uscita Z diventa 1 e rimane tale fino al primo 1 successivo su X[0] con X[1] uguale a 0, allorché l'uscita Z ritorna a 0. Esempio:

```
X[0]...0010|1100|0010|1000|1101|1001|1...
X[1]...0000|0000|0000|1010|0001|0011|1...
Z    ...0000|0000|0000|1111|0001|0001|1...
```

Tracciare il diagramma di temporizzazione [4/9 punti] come verifica della correttezza dell'unità. Nota: si puo' svolgere l'esercizio su carta oppure con ausilio del simulatore salvando una copia dell'output (diagramma temporale) e del programma Verilog su USB-drive del docente. Modello del diagramma temporale da tracciare:



Testbench:

```
module Testbench;
  reg reset_; initial begin reset_=0; #7 reset_=1; #300; $stop; end
  reg clock; initial clock=0; always #5 clock<=(!clock);
  reg [1:0] X;
  wire[2:0] STAR=Xxx.STAR;
  initial begin X='B00;
    wait(reset_==1);
    @(posedge clock); X<='B00;@(posedge clock); X<='B00;@(posedge clock); X<='B01;@(posedge clock); X<='B00;
    @(posedge clock); X<='B01;@(posedge clock); X<='B01;@(posedge clock); X<='B00;@(posedge clock); X<='B00;
    @(posedge clock); X<='B00;@(posedge clock); X<='B00;@(posedge clock); X<='B01;@(posedge clock); X<='B00;
    @(posedge clock); X<='B11;@(posedge clock); X<='B00;@(posedge clock); X<='B10;@(posedge clock); X<='B00;
    @(posedge clock); X<='B01;@(posedge clock); X<='B01;@(posedge clock); X<='B00;@(posedge clock); X<='B11;
    @(posedge clock); X<='B01;@(posedge clock); X<='B00;@(posedge clock); X<='B10;@(posedge clock); X<='B11;
    @(posedge clock); X<='B11;#10
    $finish;
    end
  XXX Xxx(X,Z,clock,reset_);
endmodule
```

COMPITO di ARCHITETTURA DEI CALCOLATORI del 22-06-2021

SOLUZIONE

ESERCIZIO 1

```
.text
.globl main
hada1:
   mv       a6,a0      # l
   fmv.s.x  fa5,zero   # fa5=0.0
   fcvt.s.w fa4,a6     # (float)l
   fsqrt.s  fa3,fa4    # fa4=m=sqrt(l)

   mulw     a0,a6,a6   # l*l
   slli     a0,a0,2    # * sizeof(float)
   li       a7,9       # sbrk
   ecall               # a0=&H

   add      t0,x0,x0   # i=0
h1_fori_start:
   slt      t6,t0,a6   # i<?l
   beq      t6,x0,h1_fori_end

   add      t1,x0,x0   # j=0
h1_forj_start:
   # prepara &H[i,j]
   mul      t6,t0,a6   # i*l
   add      t6,t6,t1   # i*l+j
   slli     t6,t6,2    # 4*(i*l+j)
   add      a5,t6,a0   # a5=&H[i,j]

   slt      t6,t1,a6   # j<?l
   beq      t6,x0,h1_forj_end

h1_if1:
   bne      t0,x0,h1_else# i!=0: h1_else
   bne      t1,x0,h1_else# k!=0: h1_else
   li       t6,1       # t6=1
   fcvt.s.w fa1,t6     # fa1=(float)1.0
   fdiv.s   fa2,fa1,fa3# fa1=1.0/sqrt(l)

   fsw      fa2,0(a5)  # H[i,j]=(.)
   b        h1_end_else
h1_else:
   add      t2,x0,x0   # k=0
h1_fork_start:
   slti     t6,t2,11   # k<=?10
   beq      t6,x0,h1_fork_end
#---- h1_fork body_start
   li       t5,1
   sll      t5,t5,t2   # t5=t=1<<k
h1_if2:
   slt      t6,t0,t5   # i<?t
       bne      t6,x0,h1_if2_end
   slt      t6,t1,t5   # j<?t
   beq      t6,x0,h1_if2_end

   sub      t6,t0,t5   # i-t
   mul      t6,t6,a6   # (i-t)*l
   add      t6,t6,t1   # (i-t)*l+j
   slli     t6,t6,2    # 4*(.)
   add      t6,t6,a0   # &H[i-t,j]
   flw      fa2,0(t6)  # H[.]
   fsw      fa2,0(a5)  # H[i,j]=(.)
h1_if2_end:
#---- h1_fork body_end
   addi     t2,t2,1    # ++k
   b        h1_fork_start
h1_fork_end:
h1_end_else:
   addi     t1,t1,1    # ++j
   b        h1_forj_start
h1_forj_end:
   addi     t0,t0,1    # ++i
   b        h1_fori_start
h1_fori_end:
   ret                 # a0=&H

main:
   fmv.s.x fa0,zero    # s = 0
   li       a0,4       # a0=N=4
   call     hada1      # returns &H
   addi     a5,a0,64   # end of H (next-byte)
main_for:
   flw      fa4,0(a0)  # H[k}
   fadd.s   fa0,fa0,fa4 # s+=H[k]
   addi     a0,a0,4    # update k*4
   bne      a0,a5,main_for
   li       a7,2       # print_float
   ecall
   li       a7,10      # exit
   ecall
```

```
Run I/O
2.0
-- program is finished running (0) --
```

ESERCIZIO 2

Sia X il generico riferimento, A=associativita', B=dimensione del blocco, C=capacita' della cache.
Si ricava S=C/B/A=# di set della cache=64/8/2, XM=X/B, XS=XM%S, XT=XM/S.
A=2, B=8, C=64, RP=LRU, Thit=4, Tpen=40, 20 references:

```
=== T    X   XM  XT  XS  XB  H [SET]:USAGE [SET]:MODIF  [SET]:TAG
=== R  125   15   3   3   5  0 [3]:1,0 [3]:0,0  [3]:3,-
=== W  170   21   5   1   2  0 [1]:1,0 [1]:0,0  [1]:5,-
=== R  167   20   5   0   7  0 [0]:1,0 [0]:0,0  [0]:5,-
=== W  245   30   7   2   5  0 [2]:1,0 [2]:0,0  [2]:7,-
=== R  183   22   5   2   7  0 [2]:0,1 [2]:0,0  [2]:7,5
=== W  119   14   3   2   7  0 [2]:1,0 [2]:0,0  [2]:3,5
=== R  235   29   7   1   3  0 [1]:0,1 [1]:0,0  [1]:5,7
=== W  163   20   5   0   3  1 [0]:1,0 [0]:1,0  [0]:5,-
=== R  288   36   9   0   0  0 [0]:0,1 [0]:1,0  [0]:5,9
=== W  309   38   9   2   5  0 [2]:0,1 [2]:0,0  [2]:3,9
=== R  310   38   9   2   6  1 [2]:0,1 [2]:0,0  [2]:3,9
=== W  308   38   9   2   4  1 [2]:0,1 [2]:0,1  [2]:3,9
=== R  213   26   6   2   5  0 [2]:1,0 [2]:0,1  [2]:6,9
=== W  196   24   6   0   4  0 [0]:1,0 [0]:0,0  [0]:6,9
=== R  377   47  11   3   1  0 [3]:0,1 [3]:0,0  [3]:3,11
=== W  166   20   5   0   6  0 [0]:0,1 [0]:0,0  [0]:6,5
=== R  362   45  11   1   2  0 [1]:1,0 [1]:0,0  [1]:11,7
=== W  233   29   7   1   1  1 [1]:0,1 [1]:0,1  [1]:11,7
=== R  163   20   5   0   3  1 [0]:0,1 [0]:0,0  [0]:6,5
=== W  169   21   5   1   1  0 [1]:1,0 [1]:0,1  [1]:5,7
-------------------------------------
P1 Nmiss=15   Nhit=5  Nref=20  mrate=0.750000   AMAT=th+mrate*tpen=34
```

LISTA BLOCCHI USCENTI:

```
(out: XM=30 XT=7  XS=2 )
(out: XM=22 XT=5  XS=2 )
(out: XM=14 XT=3  XS=2 )
(out: XM=20 XT=5  XS=0 )
(out: XM=36 XT=9  XS=0 )
(out: XM=21 XT=5  XS=1 )
(out: XM=45 XT=11 XS=1 )
```

CONTENUTI dei SET al termine

ESERCIZIO 3

Il modo 4 viene utilizzato per realizzare sull'uscita OUT (es. OUT2 per CR2) un impulso da "1" verso "0" dopo un ritardo temporale pari a N/fc essendo N la costante di tempo scritta nel registro di conteggio (es. CR2), mentre fc è la frequenza applicata sul piedino CLK corrispondente al contatore di interesse (es. CLK2 per CR2).



i) In figura sono rappresentati i segnali appena discussi (OUT, GATE, CLK); inoltre, "Counter" (CR2) indica il valore assunto dal contatore durante il conteggio, mentre CWR indica il valore impostato nel registro CWR e /WR è il segnale di scrittura applicato per poter scrivere nei registri CR2 e CWR.

ii) La parola di controllo deve valere 1010'1000=0xA8, essendo necessario sufficiente una scrittura del byte più significativo (MSB) per scrivere i 16 bit della costante N=64000=0xFA00, ovvero basta scrivere 0xA8 in CRW e in CR2 (il byte meno significativo si assume re-impostato automaticamente a zero).

COMPITO di ARCHITETTURA DEI CALCOLATORI del 22-06-2021

SOLUZIONE

ESERCIZIO 4

Dopo aver ricevuto 3 volte X[1], quando X=11 avrò Z=1 e quando X=01 torno con Z00.

Diagramma degli stati:



Codice Verilog del modulo da realizzare (possibile soluzione con Mealy-Ritardato):

```
module XXX(x,z,clock,reset_);
input        clock,reset_;
input  [1:0] x;
output       z;
reg          OUTR;
reg    [2:0] STAR;
parameter S0='B000,S1='B001,S2='B010,S3='B011,S4='B100;
always @(reset_==0) #1 begin STAR<=S0; OUTR<=0; end
assign z=OUTR;

always @(posedge clock) if(reset_==1) #3
casex(STAR)
   S0:begin STAR<=(x[0]==1)?S1:S0; OUTR<=0; end
   S1:begin STAR<=(x[0]==1)?S2:S1; OUTR<=0; end
   S2:begin STAR<=(x[0]==1)?S3:S2; OUTR<=0; end
   S3:begin STAR<=(x=='B11)?S4:S3; OUTR<=(x=='B11)?1:0; end
   S4:begin STAR<=(x=='B01)?S3:S4; OUTR<=(x=='B01)?0:1; end
endcase
endmodule
```

Diagramma di Temporizzazione: (template)